
1 REXX: Usando el fichero TEMP y copiando con IEBGENER
DES1.USRJMP.CMD(SKL2FILE)

	/* rexx                                                            */

/*******************************************************************/

/* procedimiento para cargar un eskeleto expandido en un fichero   */

/* otro ejemplo con clist es el DES1.USRJMP.CMD(extrae)            */

/*******************************************************************/

/* TRACE i */

DSN = 'DES1.USRJMP.COMPILA.JCL'

ADDRESS ISPEXEC "EDIT DATASET('"DSN"')"

/*******************************************************************/

ADDRESS ISPEXEC "FTOPEN TEMP"

ADDRESS ISPEXEC "FTINCL COMPZZSM"

ADDRESS ISPEXEC "FTCLOSE"

ADDRESS ISPEXEC "VGET (ZTEMPF)"

/*******************************************************************/

ADDRESS ISPEXEC "VGET (ZTEMPF)"

ADDRESS TSO "ALLOC F(SYSUT1)   DA('"ZTEMPF"')  SHR"

ADDRESS TSO "ALLOC F(SYSUT2)   DSNAME('"DSN"') SHR"

ADDRESS TSO "ALLOC F(SYSIN)    DUMMY SHR"

ADDRESS TSO "ALLOC F(SYSPRINT) DUMMY SHR"

ADDRESS TSO "ALLOC F(OUTDD)    DSNAME(*)"

ADDRESS TSO "IEBGENER"

ADDRESS TSO "Free Fi(SYSUT1 SYSUT2 SYSIN SYSPRINT)"

/*******************************************************************/

ADDRESS ISPEXEC "EDIT DATASET('"DSN"')"

exit




2 REXX: allocando dataset como ISPFILE

DES1.USRJMP.CMD(SKL2FIL2)
	/* rexx                                                            */

/*******************************************************************/

/* procedimiento para cargar un eskeleto expandido en un fichero   */

/* este procedimiento se basa en la clist DES1.USRJMP.CMD(EXTRAE)  */

/*******************************************************************/

   TRACE i

   dsn = 'DES1.USRJMP.COMPILA.JCL'

/*******************************************************************/

/* allocar_fichero:                                                */

   x = outtrap('On')

   address tso "delete '"dsn"'"

   x = outtrap('Off')

   ref="tracks space(1,1) lrecl(80) recfm(f b) " ,

       "dsorg(ps) blksize(23440)"

   "alloc da("dsn") f(ispfile) new catalog "ref" reuse"

/* address ispexec "edit dataset('"dsn"')"                         */

/*******************************************************************/

   address ispexec "ftopen "

   address ispexec "ftincl NEWCOPY"

   address ispexec "ftclose"

/*******************************************************************/

   address ispexec "edit dataset('"dsn"')"

   tso address 'free  fi(ispfile)'

   exit




3 Clist: Utilidad EXTRAER
Esta sencilla utilidad sirve para extraer registros de un fichero grande, saltándose el numero de registros iniciales que se indique, y copiando los n registros siguientes en el fichero de salida.            

3.1 Panel de la utilidad Extraer
	------------------------------------------------------------------------------ 

      JMP TC03                Extraer registros            USUARIO: USRJMP      

 ------------------------------------------------------------------------------ 

 Command ==>  ________________________________________________________________  
        Esta utilidad sirve para extraer registros de un fichero grande,        

        Saltándose el numero de registros iniciales que se indique, y           

        copiando los n registros siguientes en el fichero de salida.            

             Fichero Origen: __________________________________________         

             Fichero Salida: __________________________________________         

             Numero de registros a saltar : ______                              

             Numero de registros a copiar : ______                              

             Longitud del registro entrada: ____                                

              Formato del registro entrada: ___  (FB/FBA)                       

              SUBMITIR extracción ........: NO  (s/SI)                          




3.2 Panel Extraer

	)ATTR DEFAULT(%@!)

    | TYPE(OUTPUT) INTENS(LOW) CAPS(OFF) JUST(RIGHT) PAD(' ')

    { TYPE(input)  iNTENS(high) CAPS(On) just(left)  pad('_')

    } TYPE(TEXT)   INTENS(LOW) CAPS(OFF)

    [ TYPE(TEXT)   COLOR(YELLOW) CAPS(OFF)

    # TYPE(output) COLOR(red) CAPS(OFF)

    ] TYPE(TEXT)   COLOR(GREEN) CAPS(OFF)

    _ TYPE(INPUT)  INTENS(HIGH) PAD('_')

)BODY

@

%------------------------------------------------------------------------------

@     JMP#maquina@           [Extraer registros@           USUARIO:]&ZUSER  @

%------------------------------------------------------------------------------

%Command ==> _ZCMD                                                            @

@

@       Esta utilidad sirve para%extraer@registros de un fichero grande,

@       Saltandose el numero de registros iniciales que se indique, y

@       copiando los%n@registros siguientes en el fichero de salida.

@

@           ]Fichero Origen:{FileIn                                    @

%

@           ]Fichero Salida:{FileOut                                   @

%

@           ]Numero de registros a saltar :{saltar@

@           ]Numero de registros a copiar :{copiar@

@

@           ]Longitud del registro entrada:{lrec@

@           ] Formato del registro entrada:{z  @[(FB/FBA)@

@

@           ] SUBMITIR extraccion ........:{z @[(s/SI)@

@

@

)INIT

 .ZVARS  = '(recf,submitir)'

 &submitir = NO

 .cursor= FileIn

 vget(maquina,filein,Fileout,saltar,copiar,lrec,recf)

)PROC

 VER(&filein,dsname)

 VER(&fileout,dsname)

 vput (filein,fileout,saltar, copiar,lrec,recf)

)END







3.3 Clist Extraer          

	/* <<<<<<<<<<<<<<<<<<<< EXTRAER  >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>  */

/* ========= PROCEDIMIENTO PARA EXTRAER REGISTROS DE UN FICHERO ==== */

/*                                                                   */

   PROC 0 TEST

/*                                                                   */

   SET &TEST = TESTNO

   IF &TEST = TEST THEN CONTROL SYMLIST CONLIST MSG

   ELSE CONTROL NOMSG

/*                                                                   */

   SET &USUARIO = &SYSUID

   IF &USUARIO = USRJMP1 THEN SET &USUARIO = USRJMP
   SET USER = &USUARIO

/*                                                                   */

   SET &SALTAR = 0

   SET &COPIAR = 100

   ISPEXEC DISPLAY PANEL(EXTRAER)

   SET &RETORNO = &LASTCC

   IF &RETORNO EQ 8 THEN EXIT

   IF &RETORNO NE 0 THEN +

      DO

         WRITE CODIGO DE RETORNO: &RETORNO

         EXIT

      END

/*                                                                   */

   IF &TEST = TEST THEN +

      DO

         WRITE &FILEIN

         WRITE &FILEOUT

         WRITE &SALTAR

         WRITE &SACAR

      END

/*                                                                   */

   IF &FILEIN = &FILEOUT THEN +

      DO

         WRITE NO PUEDEN COINCIDIR FICHERO DE ENTRADA Y DE SALIDA

         EXIT

      END

/*                                                                   */

   IF &LASTCC NE 0 THEN +

      DO

         WRITE CODIGO RETORNO &LASTCC

         READ SEGUIR

         EXIT

      END

/*                                                                   */

   SET &DSN = '&SYSUID..SALIDA.TMP'

   IF &SYSDSN(&DSN) = OK THEN DELETE &DSN

   IF &TEST = TEST THEN  DO

      WRITE BORRAR &DSN : RETORNO &LASTCC

      WRITE ALLOC RETORNO &LASTCC

   END

/*                                                                   */

   SET JCLNAME  = &USER.E

   IF &TEST = TEST THEN WRITE &JCLNAME

/*                                                                   */

   SET &ESKELETO = EXTRAER

   IF &TEST = TEST THEN WRITE &ESKELETO

/*                                                                   */

   FREE FI(ISPFILE)

   IF &TEST = TEST THEN  DO

      WRITE FREE RETORNO &LASTCC

   END

/* ALLOCATE FI(ISPFILE) DA(&DSN) OLD REUSE                           */

/*                                                                   */

   ALLOC FILE(ISPFILE) DATASET(&DSN) NEW CATALOG +

         UNIT(3380) RECFM(F B) LRECL(80) +

         BLKSIZE(23440) TRACKS SPACE(3 1) REUSE

/* ----------------------------------------------------------------- */

/* ALLOCA UN DATASET, NO EL FICHERO TEMPORAL DE ISPF PARA EL ER      */

/* ----------------------------------------------------------------- */

/*                                                                   */

/*                                                                   */

   ISPEXEC FTOPEN
   IF &TEST = TEST THEN  DO

      WRITE FTOPEN RETORNO &LASTCC

   END

/*                                                                   */

   IF &LASTCC NE 0 THEN GOTO ERROR1

/*                                                                   */

   ISPEXEC FTINCL &ESKELETO
   IF &LASTCC NE 0 THEN GOTO ERROR2

/*                                                                   */

   ISPEXEC FTCLOSE
   IF &LASTCC NE 0 THEN GOTO ERROR3

/*                                                                   */

/* WRITENR &CREAR ' - ' &DSN                                         */

/* READ  SEGUIR                                                      */

   IF &SUBMITIR = SI +

   OR &SUBMITIR = S THEN DO

      SUBMIT &DSN
      WRITENR PROCESO SUBMITIDO OK

   END

/*                                                                   */

   GOTO EXIT

/* -------------------------------------- */

ERROR1: ISPEXEC SETMSG MSG(JMPM001X)

        GOTO EXIT

ERROR2: ISPEXEC SETMSG MSG(JMPM002X)

        ISPEXEC FTCLOSE

        GOTO EXIT

ERROR3: ISPEXEC SETMSG MSG(DDSM003X)

/* -------------------------------------- */

EXIT: FREE FI(ISPFILE)

EXIT

END




3.4 Esqueleto Extraer

	//&USER.E JOB (DE0,MSM),

//        CLASS=A,

//        MSGCLASS=X,

//        REGION=4M,

//        NOTIFY=&USER
//*

//LIBPROC JCLLIB ORDER=DES1.PROCS

//***----------------------------------------------------------*****

//***  extraer &copiar registros empezando en el registro &saltar
//***----------------------------------------------------------*****

//R100     EXEC PGM=IDCAMS

//SYSPRINT DD SYSOUT=A

//SYSIN DD *

   DEL &FILEOUT
   SET MAXCC=0

/*

//***----------------------------------------------------------*****

//***  extraer &copiar registros empezando en el registro &saltar
//***----------------------------------------------------------*****

//R100     EXEC PGM=IDCAMS

//SYSPRINT DD SYSOUT=A

//INPUT    DD DSN=&FILEIN,

//            DISP=SHR

//OUTPUT   DD DSN=&FILEOUT,

//            DISP=(NEW,CATLG,CATLG),

//            LRECL=&LREC,

//            RECFM=&RECF,

//            SPACE=(CYL,(100,1),RLSE)

//SYSIN DD *

   REPRO -

     INFILE(INPUT)   -

     OUTFILE(OUTPUT) -

     SKIP(&SALTAR.)  -

     COUNT(&COPIAR.)

/*

